MATHEMATICA

Geometría - Circunferencia

Ricardo Villafaña Figueroa

Contenido

Ecuación de la circunferencia dados las coordenadas de su centro y su radio

Ejemplo

Encontrar la ecuación de la circunferencia con centro en A (-1, 5) y radio 7.

Solución

Definir las coordenadas del centro y el radio:

```
centro = {-1, 5};
h = -1;
k = 5;
radio = 7;
```

Calcular la ecuación de la circunferencia

```
circunferencia = (x - h)^2 + (y - k)^2 == radio^2 // Expand
```

$26 + 2x + x^2 - 10y + y^2 == 49$

Graficar la circunferencia y el centro:

```
ContourPlot[Evaluate[circunferencia],
 {x, -10, 10}, {y, -5, 15},
 Epilog →
  {Red,
   PointSize[Medium],
   Point[centro]},
 Frame → False,
 Axes → True,
 GridLines → Automatic,
 ImageSize → 350]
```



Ejemplo de Manipulación

```
Manipulate[
 ContourPlot[Evaluate[(x - h)^2 + (y - k)^2 == radio^2],
  {x, -10, 10}, {y, -10, 10},
  Epilog →
   {Red, PointSize[Medium],
    Point[{h, k}]},
  Frame → False,
  Axes → True,
  GridLines → Automatic],
 {{h, 0}, -5, 5, Appearance → "Labeled"},
 {{k, 0}, -5, 5, Appearance → "Labeled"},
 {{radio, 4}, 1, 8, Appearance → "Labeled"}]
```



La ecuación de la circunferencia dados tres puntos

Ejemplo

Encontrar la ecuación de la circunferencia que pasa por la intersección de las siguientes tres rectas:

```
eq1 = 4 x - 5 y + 8 == 0;
eq2 = 6 x - y - 14 == 0;
eq3 = 2 x + 4 y + 4 == 0;
```

Calcular el punto de intersección entre la recta 1 y la recta 2:

```
sol1 = Solve[{eq1, eq2}, {x, y}]

{{x → 3, y → 4}}
```

```
punto1 = {x, y} /. sol1[[1]]

{3, 4}
```

Calcular el punto de intersección entre la recta 2 y la recta 3:

```
sol2 = Solve[{eq2, eq3}, {x, y}]

{{x → 2, y → -2}}
```

```
punto2 = {x, y} /. sol2[[1]]

{2, -2}
```

Calcular el punto de intersección entre la recta 1 y la recta 3:

```
sol3 = Solve[{eq1, eq3}, {x, y}]
```

```
{{x → -2, y → 0}}
```

```
punto3 = {x, y} /. sol3[[1]]
```

```
{-2, 0}
```

Con los tres puntos de intersección obtenidos, calculamos los valores de D, E y F de la circunferencia.

Definición general de la circunferencia:

```
circunferencia = x^2 + y^2 + D x + E1 y + F == 0
```

```
F + D x + x^2 + E1 y + y^2 == 0
```

Sustituir el primer punto:

```
c1 = ReplaceAll[circunferencia, sol1]
```

```
{25 + 3 D + 4 E1 + F == 0}
```

```
c1 = c1[[1]]
```

```
25 + 3 D + 4 E1 + F == 0
```

Sustituir el segundo punto:

```
c2 = c2[[1]]
```

```
8 + 2 D - 2 E1 + F == 0
```

Sustituir el tercer punto:

```
c3 = ReplaceAll[circunferencia, sol3]
```

$\{4 - 2D + F == 0\}$

```
c3 = c3[[1]]
```

$4 - 2D + F == 0$

Resolver el sistema de ecuaciones formado:

```
sol4 = Solve[{c1, c2, c3}, {D, E1, F}]
```

$\left\{\left\{D \to -\frac{29}{13}, E1 \to -\frac{32}{13}, F \to -\frac{110}{13}\right\}\right\}$

Reemplazar en la ecuación general de la circunferencia el valor de las variables obtenidas:

```
c4 = ReplaceAll[circunferencia, sol4]
```

$\left\{-\frac{110}{13} - \frac{29\,x}{13} + x^2 - \frac{32\,y}{13} + y^2 == 0\right\}$

Graficar las tres líneas, la circunferencia y los tres puntos:

```
ContourPlot[Evaluate[{eq1, eq2, eq3, c4}], {x, -10, 10},
 {y, -10, 10},
 Epilog →
  {Red, PointSize[Medium],
   Point[punto1], Point[punto2], Point[punto3]},
 Frame → False,
 Axes → True,
 GridLines → Automatic]
```



La ecuación de la circunferencia dado dos puntos extremos de su diámetro

Ejemplo

Encontrar la ecuación de la circunferencia cuyos puntos finales de su diámetro son A(0,0), B (5, 0). Encontrar las coordenadas de su centro, la longitud del radio, dibujar la gráfica de la circunferencia dada y calcular su área.

Solución

Definir los puntos:

```
A = {0, 0};
B = {5, 0};
```

Calcular el punto medio:

```
puntoMedio = (A + B) / 2
```

$$\left\{\frac{5}{2}, 0\right\}$$

Calcular el radio:

```
radio = EuclideanDistance[A, puntoMedio]
```

$$\frac{5}{2}$$

Calcular la ecuación de la circunferencia:

```
h = puntoMedio[[1]];
k = puntoMedio[[2]];
circunferencia =
  (x - h)^2 + (y - k)^2 - radio^2 == 0
```

$$-\frac{25}{4} + \left(-\frac{5}{2} + x\right)^2 + y^2 == 0$$

```
% // Expand
```

$-5x + x^2 + y^2 == 0$

Graficar la circunferencia, los puntos extremos del diámetro, el centro y el diámetro:

```
ContourPlot[Evaluate[circunferencia],
 {x, 0, 5}, {y, -3, 3},
 Epilog ->
  {Red, PointSize[Medium],
   Point[A], Point[B], Point[puntoMedio],
   Blue,
   Line[{A, B}]},
 AspectRatio → Automatic,
 Frame → False,
 GridLines → Automatic
]
```



Intersección de recta y circunferencia: tangentes a la circunferencia

Ejemplo

Determinar las coordenadas de los puntos donde la recta $y - 2 = 0$ interseca a la circunferencia $x^2 + y^2 - 4x - 8y - 16 = 0$.

Solución

```
circunferencia = x^2 + y^2 - 4 x - 8 y - 16 == 0
```

$-16 - 4x + x^2 - 8y + y^2 == 0$

```
recta = y - 2 == 0
```

$-2 + y == 0$

```
soluciones = Solve[{circunferencia, recta}, {x, y}] //
  Expand
```

$\{\{x \to 2 - 4\sqrt{2}, y \to 2\}, \{x \to 2 + 4\sqrt{2}, y \to 2\}\}$

```
p1 = {x, y} /. soluciones[[1]]
```

$\{2 - 4\sqrt{2}, 2\}$

```
p2 = {x, y} /. soluciones[[2]]
```

$\{2 + 4\sqrt{2}, 2\}$

Dibujar la circunferencia, la recta y los puntos de intersección:

```
ContourPlot[Evaluate[{circunferencia, recta}],
 {x, -6, 10}, {y, -3, 12},
 Epilog ->
  {Red, PointSize[Medium],
   Point[p1], Point[p2]},
 AspectRatio → Automatic,
 Frame → False,
 Axes → Automatic,
 GridLines → Automatic,
 ImageSize → 350]
```



Ejemplo

Determinar las coordenadas de los puntos donde la recta $2x - y - 10 = 0$ interseca a la circunferencia $x^2 + y^2 = 20$.

Solución

```
circunferencia = x^2 + y^2 == 20
```
```
x^2 + y^2 == 20
```

```
recta = 2 x - y - 10 == 0
```
```
-10 + 2 x - y == 0
```

```
soluciones = Solve[{circunferencia, recta}, {x, y}]
```
```
{{x → 4, y → -2}, {x → 4, y → -2}}
```

```
p1 = {x, y} /. soluciones[[1]]
```
```
{4, -2}
```

Graficar la circunferencia, la recta y el punto de intesección.

```
ContourPlot[Evaluate[{circunferencia, recta}],
 {x, -6, 10}, {y, -6, 6},
 Epilog ->
  {Red, PointSize[Medium],
   Point[p1]},
 AspectRatio → Automatic,
 Frame → False,
 Axes → Automatic,
 GridLines → Automatic,
 ImageSize → 350]
```



Ejemplo

Determinar la ecuación de la circunferencia que es tangente a la recta $x - y + 3 = 0$ en el punto (2, 5), y el centro está sobre la recta $x + 2y - 5 = 0$.

Solución

```
rectaPuntoPendiente[A_, m_] := m (x - A[[1]]) + A[[2]]
```

```
rectaTangente = x - y + 3;
```

```
rectaCentro = x + 2 y - 5;
```

```
puntoTangencia = {2, 5};
```

Calcular la pendiente de la recta tangente:

```
a = Coefficient[rectaTangente, x];
b = Coefficient[rectaTangente, y];
c = Part[rectaTangente, 1];
```

```
pendiente = -a/b
```

```
1
```

Calcular la ecuación de la perpendicular a la tangente y que pasa por el centro:

```
rectaPerpedicular =
 rectaPuntoPendiente[puntoTangencia, -1/pendiente]
```

```
7 - x
```

Calcular la intersección entre la perpendicular y la recta que pasa por el centro:

```
solucion =
 Solve[{rectaCentro == 0, rectaPerpedicular == y},
  {x, y}]
```

```
{{x → 9, y → -2}}
```

Definir las coordenadas del centro de la circunferencia:

```
centro = {x, y} /. solucion[[1]]
```

```
{9, -2}
```

Calcular el radio de la circunferencia:

```
radio = EuclideanDistance[puntoTangencia, centro]
```

$7\sqrt{2}$

Dibujar la recta tangente, la recta que pasa por el centro de la circunferencia, el centro y la circunferencia:

```
ContourPlot[
 Evaluate[{rectaTangente == 0, rectaCentro == 0}],
 {x, -2, 20}, {y, -15, 10},
 Epilog → {
   Red, PointSize[Medium],
   Point[centro],
   Point[puntoTangencia],
   Blue,
   Circle[centro, radio]},
 AspectRatio → Automatic,
 Frame → False,
 Axes → Automatic,
 GridLines → Automatic,
 ImageSize → 300]
```



Circunferencia inscrita en un triángulo

Ejemplo

Las bisectrices de un triángulo se cortan en un punto denominado incentro (centro de la circunferencia inscrita en el triángulo).

Calcular la ecuación de la circunferencia inscrita en el triángulo cuyos vértices son los siguientes: A (-2, 1), B (4, 4), C1 (3, -1).

Solución

Definir una función para calcular la ecuación de una recta dados dos de sus puntos:

```
recta[A_, B_, x_] := (B[[2]] - A[[2]])/(B[[1]] - A[[1]]) (x - A[[1]]) + A[[2]]
```

Definir una función para calcular la distancia de un punto a una recta:

```
DistanciaPuntoRecta[a_, b_, c_, P_] :=
  Abs[a P[[1]] + b P[[2]] + c]/Sqrt[a^2 + b^2]
```

Definir una función para extraer los coeficientes de una recta:

```
coeficientes[recta_] :=
 Module[{a, b, c},
  a = Coefficient[recta, x];
  b = Coefficient[recta, y];
  c = Part[recta, 1];
  Return[{a, b, c}]]
```

Definir los tres puntos dados:

```
A = {-2, 1}; B = {4, 4}; C1 = {3, -1};
```

Calcular las ecuaciones de cada uno de los lados y calcular los coeficientes de cada una de las rectas encontradas:

```
recta1 = recta[A, B, x] - y // Expand
```

$2 + \frac{x}{2} - y$

```
{a1, b1, c1} = coeficientes[recta1]
```

$\left\{\frac{1}{2}, -1, 2\right\}$

```
recta2 = recta[B, C1, x] - y // Expand
```

$-16 + 5\,x - y$

```
{a2, b2, c2} = coeficientes[recta2]
```

$\{5, -1, -16\}$

```
recta3 = recta[C1, A, x] - y // Expand
```

$\frac{1}{5} - \frac{2\,x}{5} - y$

```
{a3, b3, c3} = coeficientes[recta3]
```

$\left\{-\frac{2}{5}, -1, \frac{1}{5}\right\}$

Dibujar el triángulo formado:

```
ContourPlot[
 Evaluate[{recta1 == 0, recta2 == 0, recta3 == 0}],
 {x, -5, 5}, {y, -5, 5},
 Frame → False,
 Axes → True,
 ImageSize → 300]
```



Calcular cada una de las ecuaciones que formarán las bisectrices:

```
P = {x, y};
eq1 = DistanciaPuntoRecta[a1, b1, c1, P] // Evaluate
```

$$\frac{2\,\text{Abs}\left[2 + \frac{x}{2} - y\right]}{\sqrt{5}}$$

```
eq2 = DistanciaPuntoRecta[a2, b2, c2, P]
```

$$\frac{\text{Abs}[-16 + 5\,x - y]}{\sqrt{26}}$$

```
eq3 = DistanciaPuntoRecta[a3, b3, c3, P]
```

$$\frac{5\,\text{Abs}\left[\frac{1}{5}-\frac{2\,x}{5}-y\right]}{\sqrt{29}}$$

Calcular la primera bisectriz igualando las ecuaciones eq1 y eq2:

```
bisectriz1 = eq1 == eq2 // Expand
```

$$\frac{2\,\text{Abs}\left[2+\frac{x}{2}-y\right]}{\sqrt{5}} == \frac{\text{Abs}[-16+5\,x-y]}{\sqrt{26}}$$

Calcular la segunda bisectriz igualando las ecuaciones eq2 y eq3:

```
bisectriz2 = eq3 == eq2 // Expand
```

$$\frac{5\,\text{Abs}\left[\frac{1}{5}-\frac{2\,x}{5}-y\right]}{\sqrt{29}} == \frac{\text{Abs}[-16+5\,x-y]}{\sqrt{26}}$$

Encontrar las intersecciones entre las bisectrices encontradas (cuatro):

```
soluciones = Solve[{bisectriz1, bisectriz2}, {x, y}] // N
```

```
{{x → 1.8303, y → 1.15932}, {x → -1.80847, y → -3.60474},
 {x → -2.32575, y → 8.83158}, {x → 7.41503, y → 1.39161}}
```

Calcular las coordenadas incentro (por observación de la figura anterior escogemos el punto 1):

```
incentro = {x, y} /. soluciones[[1]] // N
```

```
{1.8303, 1.15932}
```

Calcular la longitud del radio:

```
radio = DistanciaPuntoRecta[a1, b1, c1, incentro]
```

```
1.57046
```

Graficar las tres rectas, el incentro y la circunferencia correspondiente:

```
ContourPlot[
 Evaluate[{recta1 == 0, recta2 == 0, recta3 == 0}],
 {x, -5, 5}, {y, -5, 5},
 Epilog → {
   Red, PointSize[Medium],
   Point[incentro],
   Circle[incentro, radio]},
 Frame → False,
 Axes → True]
```



Nota.

Podemos crear una función que de cálculo de distancia que tome como parámetros la ecuación de la recta y el punto:

```
distanciaPuntoRecta2[recta_, P_] :=
 Module[{a, b, c, d},
  a = Coefficient[recta, x];
  b = Coefficient[recta, y];
  c = Part[recta, 1];
  d = Abs[a P[[1]] + b P[[2]] + c] / Sqrt[a^2 + b^2];
  Return[{d}]]
```

Ejemplos de uso:

```
radio1 = distanciaPuntoRecta2[recta1, incentro]
```

```
{1.57046}
```

```
radio2 = distanciaPuntoRecta2[recta2, incentro]
```

```
{1.57046}
```

```
radio3 = distanciaPuntoRecta2[recta3, incentro]
```

```
{1.57046}
```

Circunferencia circunscrita a un triángulo

Ejemplo

Encontrar la circunferencia circunscrita al triángulo cuyos vértices son A (-2, 0), B (3, 4) y C(2, 0).

Solución

$$m[A\_, B\_] := \frac{B[[2]] - A[[2]]}{B[[1]] - A[[1]]}$$

$$rectaDosPuntos[A\_, B\_, x\_] := \frac{B[[2]] - A[[2]]}{B[[1]] - A[[1]]}(x - A[[1]]) + A[[2]]$$

```
rectaPuntoPendiente[A_, m_] := m (x - A[[1]]) + A[[2]]
```

```
puntoMedio[A_, B_] := (A + B) / 2
```

```
A = {-2, 0}; B = {3, 4}; C1 = {2, 0};
```

```
D1 = puntoMedio[A, B]
```

$$\left\{\frac{1}{2}, 2\right\}$$

```
E1 = puntoMedio[B, C1]
```

$$\left\{\frac{5}{2}, 2\right\}$$

```
mAB = m[A, B]
```

$$\frac{4}{5}$$

```
mBC = m[B, C1]
```

4

```
recta1 = rectaPuntoPendiente[D1, -1/mAB] // Expand
```

$\frac{21}{8} - \frac{5\,x}{4}$

```
recta2 = rectaPuntoPendiente[E1, -1/mBC] // Expand
```

$\frac{21}{8} - \frac{x}{4}$

```
solucion = Solve[{recta1 == y, recta2 == y}, {x, y}]
```

$\{\{x \to 0,\ y \to \frac{21}{8}\}\}$

```
centro = {x, y} /. solucion[[1]]
```

$\{0,\ \frac{21}{8}\}$

```
radio = EuclideanDistance[centro, B]
```

$\frac{\sqrt{697}}{8}$

```
ContourPlot[Evaluate[{recta1 == y, recta2 == y}],
 {x, -6, 6}, {y, -2, 8},
 Epilog ->
  {PointSize[Medium],
   Point[A], Point[B], Point[C1],
   Line[{A, B, C1}],
   Red,
   Point[centro],
   Blue,
   Circle[centro, radio]},
 AspectRatio → Automatic,
 Frame → False,
 Axes → Automatic,
 GridLines → Automatic,
 ImageSize → 400]
```



Aplicaciones

Ejemplo

Encontrar la ecuación de la circunferencia que es tangente a la recta $3\,x - 4\,y - 4 = 0$ y cuyo centro está sobre las rectas $5\,x - y + 7 = 0$ y $x - 4\,y + 9 = 0$. Graficar la circunferencia encontrada y las tres rectas.

Primera solución

Encontrar el punto de intersección entre las rectas dadas:

```
Solucion = Solve[{5 x - y + 7 == 0,  x - 4 y + 9 == 0}, {x, y}]

{{x → -1, y → 2}}
```

Sustituir los valores encontrados de X e Y directamente en las coordenadas del centro de la circunferencia:

```
h = -1; k = 2;
```

Almacenar los valores de h y k en la variable centro para su graficación:

```
centro = {h, k}

{-1, 2}
```

Para calcular la longitud del radio, utilizaremos la fórmula para calcular la distancia de un punto a una recta:

$$\text{Abs}\left[\frac{A\,x1 + B\,y1 + C}{\sqrt{A^2 + B^2}}\right]$$

Ecuación de la recta tangente:

```
3 x - 4 y - 4 == 0;
```

Valores de los A, B y C de la ecuación tangente:

```
A = 3; B = -4; C1 = -4;
```

Coordenadas del centro:

```
h = -1; k = 2;
```

Calculando la longitud del radio sustituyendo en la fórmula de distancia:

$$\text{radio} = \text{Abs}\left[\frac{(A)\ (h) + (B)\ (k) + C1}{\sqrt{A^2 + B^2}}\right]$$

3

Calculando la ecuación de la circunferencia:

$$\text{circunferencia} = (x - h)^2 + (y - k)^2 - \text{radio}^2 == 0$$

$$-9 + (1 + x)^2 + (-2 + y)^2 == 0$$

Graficando las tres rectas dadas, el centro y la circunferencia:

```
ContourPlot[
 {(x - h)^2 + (y - k)^2 - radio^2 == 0, 5 x - y + 7 == 0,
  x - 4 y + 9 == 0, 3 x - 4 y - 4 == 0}, {x, -6, 6},
 {y, -6, 6}, Epilog →
  {Red, PointSize[Medium],
   Point[centro]}]
```



Segunda solución

Usar variables intermedias y crear una plantilla de solución. Con esta técnica podemos cambiar las ecuaciones dadas y observar los cambios en la gráfica.

```
linea1 = 5 x - y + 7;
linea2 = x - 4 y + 9;
```

Resolver el sistema de ecuaciones para encontrar el punto de intersección:

```
solucion = Solve[{linea1 == 0, linea2 == 0}, {x, y}]

{{x → -1, y → 2}}
```

Almacenar los valores encontrados para X e Y en las variables h y k, respectivamente:

```
{h, k} = {x, y} /. solucion[[1]]

{-1, 2}
```

Definir el centro con las coordenadas obtenidas:

```
centro = {h, k}

{-1, 2}
```

Calcular la distancia del centro a la recta tangente.

Definir la recta tangente:

```
linea3 = 3 x - 4 y - 4;
```

Aislar los coeficientes:

```
A = Coefficient[linea3, x];
B = Coefficient[linea3, y];
C1 = Part[linea3, 1];
```

Calcular la longitud del radio:

$$\text{radio} = \text{Abs}\left[\frac{A\,h + B\,k + C1}{\sqrt{A^2 + B^2}}\right]$$

3

Calcular la ecuación de la circunferencia con los valores encontrados

$$\text{circunferencia} = (x - h)^2 + (y - k)^2 - \text{radio}^2$$

$$-9 + (1 + x)^2 + (-2 + y)^2$$

Dibujar *implícitamente* la circunferencia y las tres rectas:

```
ContourPlot[
 Evaluate[{circunferencia == 0, linea1 == 0, linea2 == 0,
   linea3 == 0}], {x, -6, 6}, {y, -6, 6}, Epilog →
  {Red, PointSize[Medium],
   Point[centro]},
 Frame → False,
 Axes → True,
 GridLines → Automatic,
 ImageSize → 350]
```



Ejemplo

Deducir una(s) ecuación(es) del o de los círculos de radio 4, cuyo centro está en la recta $4\,x + 3\,y + 7 = 0$ y es o son tangentes a $3\,x + 4\,y + 34 = 0$.

Primera solución

(1) El radio dado:

```
radio = 4;
```

(2) Un punto cualquiera (h, k) que pase por la circunferencia debe también satisfacer a la recta que pasa por el centro:

```
4 x + 3 y + 7 == 0;
```

```
4 h + 3 k + 7 == 0;
```

(3) La tercera condición viene dada por la distancia que existe entre el radio y la recta tangente dada. Definimos primero la recta tangente en función de ***(h, k):***

```
3 x + 4 y + 34 == 0;
```

```
3 h + 4 k + 34 == 0;
```

Y luego definimos la ecuación de la distancia en función de la línea obtenida el radio dado:

$$\text{Abs}\left[\frac{(3)\ (h) + (4)\ (k) + 34}{\sqrt{3^2 + 4^2}}\right] == \text{radio}$$

$$\frac{1}{5}\,\text{Abs}[34 + 3\,h + 4\,k] == 4$$

Resolvemos el par de ecuaciones encontradas con la función ***solve***:

$$\text{Solve}\left[\left\{4\,h+3\,k+7==0,\ \frac{1}{5}\,\text{Abs}[34+3\,h+4\,k]==4\right\},\ \{h,\ k\}\right]$$

$$\left\{\{h\to 2,\ k\to -5\},\ \left\{h\to \frac{134}{7},\ k\to -\frac{195}{7}\right\}\right\}$$

El par de ecuaciones nos da dos pares de valores para encontrar las ecuaciones de los círculos:

Primer círculo:

$$h1=2;\ k1=-5;\ \text{radio}=4;$$
$$\text{circunferencia1}=(x-h1)^2+(y-k1)^2-\text{radio}^2==0$$

$$-16+(-2+x)^2+(5+y)^2==0$$

Segundo círculo:

$$h2=\frac{134}{7};\ k2=-\frac{195}{7};\ \text{radio}=4;$$
$$\text{circunferencia2}=(x-h2)^2+(y-k2)^2-\text{radio}^2==0$$

$$-16+\left(-\frac{134}{7}+x\right)^2+\left(\frac{195}{7}+y\right)^2==0$$

Graficamos las circunferencias encontradas, la línea que pasa por el centro y la línea tangente.

```
ContourPlot[{-16 + (-2 + x)^2 + (5 + y)^2 == 0,
   -16 + (-134/7 + x)^2 + (195/7 + y)^2 == 0, 4 x + 3 y + 7 == 0, 3 x + 4 y + 34 == 0},
 {x, -5, 25}, {y, -35, 5},
 AspectRatio → Automatic,
 Epilog →
  {Red, PointSize[Medium],
   Point[punto1], Point[punto2]}]
```



Segunda solución

```
radio = 4
```

```
4
```

```
linea1 = 4 x + 3 y + 7;
linea2 = 3 x + 4 y + 34;
```

```
l1 = ReplaceAll[linea1, {x → h, y → k}]
```

```
7 + 4 h + 3 k
```

```
l2 = ReplaceAll[linea2, {x → h, y → k}]
```

```
34 + 3 h + 4 k
```

```
A = Coefficient[l2, h];
B = Coefficient[l2, k];
C1 = Part[l2, 1];
```

$$\text{distancia} = \text{Abs}\left[\frac{(A)\ (h) + (B)\ (k) + C1}{\sqrt{A^2 + B^2}}\right]$$

$$\frac{1}{5}\,\text{Abs}[34 + 3\,h + 4\,k]$$

```
solucion = Solve[{l1 == 0, distancia == radio}, {h, k}]
```

$$\left\{\{h \to 2,\ k \to -5\},\ \left\{h \to \frac{134}{7},\ k \to -\frac{195}{7}\right\}\right\}$$

```
{h1, k1} = {h, k} /. solucion[[1]]
```

```
{2, -5}
```

$$\text{circunferencia1} = (x - h1)^2 + (y - k1)^2 - \text{radio}^2 == 0$$

$$-16 + (-2 + x)^2 + (5 + y)^2 == 0$$

```
{h2, k2} = {h, k} /. solucion[[2]]
```

$$\left\{\frac{134}{7}, -\frac{195}{7}\right\}$$

```
circunferencia2 = (x - h2)^2 + (y - k2)^2 - radio^2 == 0
```

$$-16 + \left(-\frac{134}{7} + x\right)^2 + \left(\frac{195}{7} + y\right)^2 == 0$$

```
punto1 = {h1, k1};
punto2 = {h2, k2};
```

```
ContourPlot[
 Evaluate[{circunferencia1, circunferencia2,
   linea1, linea2}],
 {x, -5, 25}, {y, -35, 5},
 AspectRatio → Automatic,
 Epilog →
  {Red, PointSize[Medium],
   Point[punto1], Point[punto2]},
 Frame → False,
 Axes → True,
 ImageSize → 350]
```



Ejemplo

Demostrar que las longitudes de las dos tangentes trazadas a una circunferencia desde un punto exterior son iguales.

Considerar la circunferencia $x^2 + y^2 = 4$, en el punto exterior (5, 0).

Solución

Solución encontrada con Geometry Expressions:



Solución encontrada con Mathematica:

Definiendo una función para calcular la pendiente entre dos puntos dados:

$$m[A\_, B\_] := \frac{B[[2]] - A[[2]]}{B[[1]] - A[[1]]}$$

Definiendo la función para calcular la ecuación de una recta dados dos de sus puntos:

$$rectaDosPuntos[A\_, B\_] := \frac{B[[2]] - A[[2]]}{B[[1]] - A[[1]]} (x - A[[1]]) + A[[2]]$$

Definiendo la función para calcular la ecuación de una recta dados un punto y su pendiente:

```
rectaPuntoPendiente[A_, m_] := m (x - A[[1]]) + A[[2]]
```

Definiendo los valores dados:

```
A = {0, 0};
C1 = {5, 0};
D1 = {x1, y1};
E1 = {x2, y2};
c = x^2 + y^2 - 4;
```

Encontrar la ecuación de la línea D1C1.

Primera condición:

La recta es perpendicular a la recta AD1:

```
mAD = m[A, D1]
```

$$\frac{y1}{x1}$$

```
rectaDC = rectaPuntoPendiente[D1, -1/mAD]
```

$$-\frac{(x - x1)\,x1}{y1} + y1$$

Segunda condición:

La recta pasa por el punto C1:

```
rectaDC2 = rectaDosPuntos[D1, C1]
```

$$y1 - \frac{(x - x1)\,y1}{5 - x1}$$

Igualamos las ecuaciones para encontrar una ecuación única:

```
eq1 = rectaDC == rectaDC2
```

$$-\frac{(x - x1)\,x1}{y1} + y1 == y1 - \frac{(x - x1)\,y1}{5 - x1}$$

Tercera condición:

La recta encontrada debe ser tangente a la circunferencia dada:

```
c1 = c /. {x → x1, y -> y1}
```

$$-4 + x1^2 + y1^2$$

Resolvemos el sistema de ecuaciones obtenido:

```
par = Solve[{eq1, c1 == 0}, {x1, y1}]
```

$$\left\{\left\{y1 \to -\frac{2\sqrt{21}}{5},\ x1 \to \frac{4}{5}\right\},\ \left\{y1 \to \frac{2\sqrt{21}}{5},\ x1 \to \frac{4}{5}\right\},\ \left\{y1 \to -\sqrt{4-x^2},\ x1 \to x\right\},\ \left\{y1 \to \sqrt{4-x^2},\ x1 \to x\right\}\right\}$$

La solución nos muestra que hay dos líneas rectas que cumplen las condiciones anteriores.

Primera línea:

```
D1 = {x1, y1} /. par[[2]]
```

$$\left\{\frac{4}{5}, \frac{2\sqrt{21}}{5}\right\}$$

```
rectaDC1 = rectaDosPuntos[D1, C1] == y // Simplify
```

$$-\frac{2(-5+x)}{\sqrt{21}} == y$$

Segunda línea:

```
D2 = {x1, y1} /. par[[1]]
```

$$\left\{\frac{4}{5}, -\frac{2\sqrt{21}}{5}\right\}$$

```
rectaDC2 = rectaDosPuntos[D2, C1] == y // Simplify
```

$$\frac{2(-5+x)}{\sqrt{21}} == y$$

Calculando la primera distancia:

```
distancia1 = EuclideanDistance[D1, C1] // Simplify
```

$$\sqrt{21}$$

Calculando la segunda distancia:

```
distancia2 = EuclideanDistance[D2, C1] // Simplify
```

$\sqrt{21}$

Graficando la circunferencia, las dos rectas tangentes encontradas, los puntos de tangencia y el punto exterior a la circunferencia:

```
ContourPlot[Evaluate[{c == 0, rectaDC1, rectaDC2}],
 {x, -6, 6}, {y, -6, 6},
 Epilog →
  {Red, PointSize[Medium],
   Point[D1], Point[D2], Point[C1]},
 Frame → False,
 Axes → True,
 GridLines → Automatic,
 ImageSize → 350]
```